NOAA, NATIONAL WEATHER SERVICE, CENTRAL REGION

COMPUTER PROGRAMS AND PROBLEMS

NWS CRCP - No. 10

A THERMODYNAMICS LIBRARY FOR AFOS PROGRAMMERS (THERMO.LB)

Warren E. Sunkel
National Weather Service Forecast Office
Topeka, Kansas

November 1983

U.S. DEPARTMENT OF COMMERCE / National Oceanic and Atmospheric Administration / National Weather Service

NOAA, National Weather Service
Central Region Computer Programs and Problems Series

NWS CRCP No. 1 A Computer Program for Determining and Displaying the Geostrophic Winds on Lakes Superior and Michigan. R. Somrek. June 1980 (PB80 225675)

NWS CRCP No. 2 Program ALEMBIC. W. Sunkel. March 1982 (PB83 114488)

NWS CRCP No. 3 A Subroutine to Find and Extract Data from an AFOS Database Product. R. Somrek. May 1982 (PB83 118828)

NWS CRCP No. 4 Programs SAVALL and STORALL. W. Sunkel. July 1982. (PB83 118448)

NWS CRCP No. 5 Program ANALYZ. T. Schwein. Sept. 1982

NWS CRCP No. 6 A Regional Weather Depiction Plot. W. Sunkel. October 1982. (PB83 194415)

NWS CRCP No. 7 The Topeka Library (TOP.LB). W. Sunkel. March 1983

NWS CRCP No. 8 A Multipurpose Weather Roundup Program. Warren E. Sunkel May 1983. (PB83 222612)

NWS CRCP No. 9 Assembly Language Graphics Library (EGR1.LB) with Fortran Interfacing. Thomas J. Egger July 1983 (PB83 239624)

NWS CRCP No. 10 A Thermodynamics Library for AFOS Programmers (THERMO.LB) Warren E. Sunkel, November 1983

NWS CRCP - No. 10

A THERMODYNAMICS LIBRARY FOR AFOS PROGRAMERS (THERMO.LB)

Warren E. Sunkel
National Weather Service Forecast Office
Topeka, Kansas

November 1983

UNITED STATES
DEPARTMENT OF COMMERCE
Malcolm Baldrige, Secretary

National Oceanic and
Atmospheric Administration
John V Byrne, Administrator

National Weather
Service
Richard E Hallgren,
Assistant Administrator

CONTENTS

A THERMODYNAMICS LIBRARY FOR AFOS PROGRAMMERS (THERMO.LB)

prepared by

Warren E. Sunkel

for

Scientific Services Division
National Weather Service Central Region
Kansas City, Missouri

INTRODUCTION

THERMO.LB is a collection of FORTRAN function subprograms which compute frequently used variables in thermodynamic analysis. The subprograms are designed for maximum computational efficiency while maintaining a high degree of scientific accuracy.

All of the subprograms contained in this library are compatible with the Data General S/230 minicomputers of the AFOS system when loaded with the FORTRAN library (FORT.LB). Each subprogram requires fewer than 300 octal locations of NREL memory, but additional memory is required by the FORT.LB supporting routines. Essentially, these subprograms add little additional memory requirement to a program which already uses floating point mathematics. THERMO.LB requires nine RDOS blocks of disk storage.

The FORTRAN coding of the routines in this library was developed by Hermann B. Wobus while he was affiliated with the now disestablished Naval Weather Research Facility in Norfolk, Virginia. This document presents the subprograms for use in AFOS applications programs requiring thermodynamic computations. The theoretical derivations of the algorithms are described by Doswell, et al. (1982).

FUNCTIONAL DESCRIPTIONS OF THE SUBPROGRAMS

Each subprogram of the Thermodynamics Library is presented individually in this document. A summary of the subprogram's design is followed by a source listing.

The number of normally relocatable (NREL) locations required by the subprogram is listed in the heading after "Memory locations." No page zero (ZREL) memory is required by these routines, and the required stack memory is unknown.

All of the subprograms for this AFOS library were compiled using Data General FORTRAN, Revision 05.57. The FORTRAN source files are sufficiently general to work equally well on most FORTRAN compilers.

CONVENTIONS AND APPROXIMATIONS

All of the subprograms contained on this library require input in degrees Celcius for temperature, potential temperature and dewpoint; and in millibars for pressure and vapor pressure. Some of the routines will also accept temperatures in degrees Kelvin (see individual documents).

Output is in units of degrees Celcius for temperature, millibars for pressure and vapor pressure, and grams per kilogram for mixing ratio.

Wobus developed his original regression equations to eight digits of accuracy. Even though only six digits are significant in the single precision floating point mathematics of the S/230, the following FORTRAN source programs retain the eight-digit constants for the purpose of documentation.

SAMPLE RUN FOR TESTING

Appendices 1 and 2 contain sample programs which use the routines from this library. Although necessarily elementary, the sample programs demonstrate practical applications of the library subprograms.

Stone (1983) presents an excellent study of these subprograms applied to the thermodynamic analysis of the physical processes of the atmosphere in an operational environment.

ACKNOWLEDGEMENTS

The author wishes to thank Chuck Doswell, Rich McNulty, Joe Schaefer, and Bob van Haaren for their contributions and suggestions.

REFERENCES

Doswell, C. A., J. T. Schaefer, D. W. McCann, T. W. Schlatter, and H. B. Wobus, 1982: Thermodynamic Analysis Procedures at the National Severe Storms Forecast Center, PREPRINTS, 9th CONFERENCE ON WEATHER FORECASTING AND ANALYSIS, American Meteorological Society, Seattle, 304-309.

Stone, Hugh M., 1983: Stability Analysis Program, NOAA EASTERN REGION COMPUTER PROGRAMS AND PROBLEMS NWS ERCP No. 9, National Weather Service, Garden City, New York.

THERMODYNAMICS LIBRARY

I. IDENTIFICATION

Module name:	TCONOF
Date:	September 22, 1983
Function:	Compute condensation temperature
Language:	FORTRAN
Memory locations:	130

II. FUNCTIONAL SUMMARY

TCONOF is a real function subprogram to compute the condensation temperature (degrees C) given the temperature and dewpoint (like units, degrees C or K) of a parcel at any pressure.

III. ENTRY POINTS

TCONO

IV. CALLING METHOD

```
TC=TCONOF(TEMP,DEWPT)
```

V. INPUT

TEMP - A single precision floating point variable or constant containing temperature in degrees (C or K)

DEWPT - A single precision floating point variable or constant containing dew point in degrees (C or K)

VI. OUTPUT

TC - Condensation temperature of the lifted parcel (degrees C)

VII. ERROR RETURNS

NONE

VIII. REFERENCED EXTERNALS

FORT.LB
- FAD1
- FFLD1
- FFST1

FML1
FSB1
FSGN1
.FARL
.FRET
.MOVE
.I

```
      FUNCTION TCONOF(TEMP,DEWPT)
C
C     COMPUTE CONDENSATION TEMPERATURE (DEGREES CELCIUS) BY LIFTING
C     TEMP AND DEWPT ARE IN LIKE UNITS (DEGREES CELCIUS OR KELVIN)
C
      S=TEMP-DEWPT
      T=TEMP
      IF(100.0-TEMP)4,5,5
    4 T=TEMP-273.16
COMPUTE CURVE FIT IN MOST EFFICIENT MANNER
    5 DLT=S*(1.2185+0.001278*T+S*(-0.002190+11.73E-6*S-5.20E-6*T))
      TCONOF=T-DLT
      RETURN
      END
```

I. IDENTIFICATION

Module name:	SATLFT
Date:	September 22, 1983
Function:	Compute temperature of lifted parcel
Language:	FORTRAN
Memory locations:	271

II. FUNCTIONAL SUMMARY

SATLFT is a real function subprogram to compute the temperature (degrees C) of a saturated parcel being lifted along a pseudo-adiabat to a given pressure.

III. ENTRY POINTS

SATLF

IV. CALLING METHOD

T=SATLFT(THM,P)

V. INPUT

THM - A single precision floating point variable or constant containing the wet bulb potential temperature of the saturated parcel (theta M) in degrees Celcius

P - A single precision floating point variable or constant containing the pressure (millibars) to which to parcel is lifted

VI. OUTPUT

T - The temperature of the parcel (degrees Celcius) at the given pressure

VII. ERROR RETURNS

NONE

VIII. REFERENCED EXTERNALS

THERMO.LB
- WOBF

FORT.LB
- ABS.
- FAD1
- FDV1
- FFLD1
- FFST1
- FLIP1
- FML1
- FPWR1
- FSB1
- FSGN1
- .FARL
- .FCAL
- .FRET
- .MOVE
- .I

```
      FUNCTION SATLFT(THM,P)
C
C     LIFT PARCEL ALONG PSEUDO-ADIABAT
C     COMPUTES TEMPERATURE (DEGREES CELCIUS) WHERE THETA MOIST (DEGREES
C     CELCIUS) CROSSES P (MILLIBARS)
C
C     REQUIRES FUNCTION SUBPROGRAM WOBF
C
CONSIDER THE EXPONENTIAL FOR POTENTIAL TEMPERATURE AS ROCP
      ROCP=0.28571428
      IF(ABS(P-1000.0)-0.0010)100,100,200
  100 SATLFT=THM
      RETURN
  200 PWRP=(P/1000.0)**ROCP
COMPUTE TEMPERATURE OF DRY ADIABATIC LIFT FOR FIRST GUESS
      TONE=(THM+273.16)*PWRP-273.16
CONSIDER PSEUDO-ADIABAT, EW1, THROUGH TONE AT P.
COMPUTE EONE=EW1-THM
      EONE=WOBF(TONE)-WOBF(THM)
      RATE=1.0
      GO TO 330
  300 CONTINUE
CONTRIBUTION TO ITERATION IS CHANGE IN T
CORRESPONDING TO CHANGE IN E
      RATE=(TTWO-TONE)/(ETWO-EONE)
      TONE=TTWO
      EONE=ETWO
  330 CONTINUE
COMPUTE ESTIMATED SATLFT,TTWO
      TTWO=TONE-EONE*RATE
CONSIDER PSEUDO-ADIABAT, EW2, THROUGH TTWO AT P.
COMPUTE ETWO=EW2-THM
      ETWO=(TTWO+273.16)/PWRP-273.16
      ETWO=ETWO+WOBF(TTWO)-WOBF(ETWO)-THM
CORRECTION TO TTWO IS EOR
      EOR=ETWO*RATE
      IF(ABS(EOR)-0.1000)400,400,300
  400 SATLFT=TTWO-EOR
      RETURN
      END
```

THERMODYNAMICS LIBRARY

I. IDENTIFICATION

Module name:	DPTOF
Date:	September 22, 1983
Function:	Compute dewpoint from known vapor pressure
Language:	FORTRAN
Memory locations:	217

II. FUNCTIONAL SUMMARY

DPTOF is a real function subprogram to compute the dewpoint temperature (degrees Celcius) as a function of water vapor pressure (millibars).

III. ENTRY POINTS

DPTOF

IV. CALLING METHOD

```
TD=DPTOF(EW)
```

V. INPUT

EW - A single precision floating point variable or constant containing vapor pressure (millibars) with respect to liquid water

VI. OUTPUT

TD - Dewpoint temperature in degrees Celcius

VII. ERROR RETURNS

If EW is outside the range 0.21382876E-09 to 1013.0, a value of -10000.0 is returned for the dewpoint

VIII. REFERENCED EXTERNALS

THERMO.LB
    VAPFW

FORT.LB

```
ABS.
ALOG.
FAD1
FDV1
FFLD1
FFST1
FML1
FSB1
FSGN1
.FARL
.FCAL
.FRET
.MOVE
.I
```

```
      FUNCTION DPTOF(EW)
C
C     COMPUTE DEWPOINT (DEGREES CELCIUS) AS A FUNCTION OF
C     WATER VAPOR PRESSURE (MILLIBARS)
C
C     IF VAPOR PRESSURE IS OUTSIDE THE RANGE 0.21382876E-09 TO 1013.0,
C     A DEWPOINT OF -1000.0 IS RETURNED TO INDICATE THE ERROR CONDITION
C
C     REQUIRES FUNCTION SUBPROGRAM VAPFW
C
CREATE TOLERANCE TO DEGREE DESIRED
      TOL=0.00010
      IF(EW-0.21382876E-09)20,20,30
   20 DPTOF=-10000.
      RETURN
   30 IF(1013.0-EW)20,100,100
CREATE GUESS BY INVERTING TETEN'S FORMULA
  100 X=ALOG(EW/6.1078)
      BOT=17.269388-X
      DPTOF=(237.3*X)/BOT
      BOT=BOT*EW
  200 EDP=VAPFW(DPTOF)
CORRECT GUESS BY DERIVATIVE OF TEMPERATURE WITH RESPECT TO VAPOR PRES.
CALCULATED FROM INVERSE OF TETEN'S FORMULA
      DTDE=(DPTOF+273.3)/BOT
      DELT=DTDE*(EW-EDP)
      DPTOF=DPTOF+DELT
CHECK TO SEE IF ANSWER CLOSE ENOUGH, IF NOT ITERATE OVER CORRECTION
      IF(ABS(DELT)-TOL)300,300,200
CHANGE SO DEW POINT IS ALWAYS LESS THAN THE TEMP.
COMPATABILITY WITH TOL IS FORCED
  300 DPTOF=DPTOF-TOL
      RETURN
      END
```

THERMODYNAMICS LIBRARY

I. IDENTIFICATION

Module name:	WMROF
Date:	September 22, 1983
Function:	Compute mixing ratio
Language:	FORTRAN
Memory locations:	154

II. FUNCTIONAL SUMMARY

WMROF is a real function subprogram to compute mixing ratio (grams per kilogram) as a function of pressure (millibars) and dewpoint temperature (degrees C or K). It incorporates a correction for a non-ideal gas.

III. ENTRY POINTS

WMROF

IV. CALLING METHOD

W=WMROF(P,TD)

V. INPUT

P - A single precision floating point variable or constant containing the pressure (millibars) of the parcel

TD - A single precision floating point variable or constant containing the parcel's dewpoint temperature (degrees C or K)

VI. OUTPUT

W - The mixing ratio of the parcel in grams per kilogram

VII. ERROR RETURNS

NONE

VIII. REFERENCED EXTERNALS

THERMO.LB

VAPFW

```
FORT.LB
        FAD1
        FDV1
        FFLD1
        FFST1
        FLIP1
        FML1
        FSB1
        FSGN1
        .FARL
        .FCAL
        .FRET
        .MOVE
        .I
```

```
      FUNCTION WMROF(P,TD)
C
C     COMPUTE MIXING RATIO (GRAMS PER KILOGRAM) AS A FUNCTION OF
C     PRESSURE (MILLIBARS) AND DEWPOINT TEMPERATURE (DEGREES CELCIUS
C     OR KELVIN)
C
C     REQUIRES FUNCTION SUBPROGRAM VAPFW
C
      T=TD
      IF(100.-T)3,4,4
    3 T=T-273.16
CURVE FIT CORRECTION FOR NON-IDEAL GAS
    4 X=0.0200*(T-12.5+7500.0/P)
      WFW=1.+0.0000045*P+0.00140*X*X
COMPUTE ACCORDING TO STANDARD FORMULA
      FWESW=WFW*VAPFW(T)
      WMROF=621.97*(FWESW/(P-FWESW))
      RETURN
      END
```

I. IDENTIFICATION

Module name:	WOBF
Date:	September 22, 1983
Function:	Compute the Wobus function
Language:	FORTRAN
Memory locations:	271

II. FUNCTIONAL SUMMARY

WOBF is a real function subprogram to compute the equivalent heat energy of moisture (Wobus function) as a function of temperature (degrees Celcius). The Wobus function is defined as the difference between the wet bulb potential temperature of a saturated parcel at (T,p) and the wet bulb potential temperature of an absolutely dry parcel at (T,p).

III. ENTRY POINTS

WOBF

IV. CALLING METHOD

```
W=WOBF(T)
```

V. INPUT

T - A single precision floating point variable or constant containing the temperature of the parcel in degrees Celcius

VI. OUTPUT

W - The Wobus function in degrees Celcius

VII. ERROR RETURNS

NONE

VIII. REFERENCED EXTERNALS

FORT.LB
- FAD1
- FDV1

FFLD1
FFST1
FLIP1
FML1
FSB1
FSGN1
.FARL
.FRET
.I

```
      FUNCTION WOBF(T)
C
C     COMPUTE THE EQUIVALENT HEAT ENERGY OF MOISTURE (WOBUS FUNCTION)
C     FROM TEMPERATURE (DEGREES CELCIUS)
C
COMPUTE BY DOUBLE ASYMPTOTIC APPROXIMATION
CONSIDER SEPARATELY IF .GT. OR .LE. 20 DEG.
CENT. FOR ALL TEMPS...THETW=THETA-WOBF(THETA)+WOBF(TEMPCON)
CENT. FOR ALL TEMPS...THETM=THETA-WOBF(THETA)+WOBF(TEMP)
      X=T-20.0
      IF(X)10,10,20
   10 CONTINUE
CURVE FIT FOR COOL TEMPERATURE RANGE
      POL=1.000+X*(-8.8416605E-3
     A +X*(1.4714143E-4+X*(-9.6719890E-7
     B +X*(-3.2607217E-8+X*(-3.8598073E-10))))
      POL=POL*POL
      WOBF=15.130/(POL*POL)
      RETURN
   20 CONTINUE
CURVE FIT FOR WARMER TEMPERATURES
      POL=1.000+X*(3.6182989E-3
     1 +X*(-1.3603273E-5+X*(4.9618922E-7
     2 +X*(-6.1059365E-9+X*(3.9401551E-11
     3 +X*(-1.2588129E-13+X*(1.6688280E-16)))))))
      POL=POL*POL
      WOBF=29.930/(POL*POL)+0.9600*X-14.800
      RETURN
      END
```

THERMODYNAMICS LIBRARY

I. IDENTIFICATION

Module name:	VAPFW
Date:	September 22, 1983
Function:	Compute saturation vapor pressure
Language:	FORTRAN
Memory locations:	212

II. FUNCTIONAL SUMMARY

VAPFW is a real function subprogram to compute the saturation vapor pressure (millibars) with respect to liquid water at temperature T (degrees C or K). The function is accurate for temperatures between -50 and +100 C.

III. ENTRY POINTS

VAPFW

IV. CALLING METHOD

```
ES=VAPFW(T)
```

V. INPUT

T - A single precision floating point variable or constant containing temperature in degrees C or K

VI. OUTPUT

ES - The saturation vapor pressure with respect to water (millibars)

VII. ERROR RETURNS

NONE

VIII. REFERENCED EXTERNALS

FORT.LB
- FAD1
- FDV1
- FFLD1

FFST1
FLIP1
FML1
FSB1
FSGN1
.FARL
.FRET
.MOVE
.I

```
      FUNCTION VAPFW(T)
C
C     COMPUTE SATURATION VAPOR PRESSURE (MILLIBARS) WITH RESPECT
C     TO WATER AS A FUNCTION OF TEMPERATURE (DEGREES CELCIUS OR KELVIN)
C
      X=T
      IF(100.0-X)3,4,4
    3 X=X-273.16
CURVE FIT FOR RANGE -50 < T < 100
    4 POL=0.99999683E-00+X*(-0.90826951E-02+
     1 X*(0.78736169E-04+X*(-0.61117958E-06+
     2 X*(0.43884187E-08+X*(-0.29883885E-10+
     3 X*(0.21874425E-12+X*(-0.17892321E-14+
     4 X*(0.11112018E-16+X*(-0.30994571E-19)))))))))
      POL=POL*POL
      POL=POL*POL
      VAPFW=6.107800/(POL*POL)
      RETURN
      END
```

APPENDIX 2

SAMPLE2.FR 09/25/83 09:41

```
C       PROGRAM SAMPLE2
C
C       COMPUTE THE DEWPOINT GIVEN TEMPERATURE AND RELATIVE HUMIDITY
C       USING ROUTINES FROM THERMO.LB
C
        ACCEPT "ENTER TEMPERATURE IN DEGREES F:          ",TEMP
        ACCEPT "ENTER RELATIVE HUMIDITY IN PERCENT:      ",HUM
C
C       CONVERT TEMPERATURE TO DEGREES C
C
        TT=5.*(TEMP-32.)/9.
C
C       COMPUTE THE SATURATION VAPOR PRESSURE AT TT
C
        ES=VAPFW(TT)
C
C       USE RELATIVE HUMIDITIY TO COMPUTE THE ACTUAL VAPOR PRESSURE
C
        E=HUM/100.*ES
C
C       CONVERT VAPOR PRESSURE TO DEWPOINT IN DEGREES C
C
        TD=DPTOF(E)
C
C       CONVERT DEWPOINT TO DEGREES F
C
        TD=TD*1.8+32
C
C       ROUND OFF AND OUTPUT
C
        I=TD+.5
        IF(TD.LT.0)I=TD-.5
        TYPE
        TYPE "DEWPOINT =",I
        END
```

APPENDIX 3

L I S T   O F   S Y M B O L S

e	vapor pressure
es	saturation vapor pressure
ew	vapor pressure with respect to water
p	pressure
T	temperature
Tc	lifted condensation temperature
Td	dewpoint temperature
W	Wobus function = theta M - theta A
w	mixing ratio
theta	potential temperature
theta A	wet bulb potential temperature for absolutely dry conditions at (T,p)
theta M	wet bulb potential temperature for saturated conditions at (T,p)
theta w	wet bulb potential temperature

NOAA SCIENTIFIC AND TECHNICAL PUBLICATIONS

The National Oceanic and Atmospheric Administration was established as part of the Department of Commerce on October 3, 1970. The mission responsibilities of NOAA are to assess the socioeconomic impact of natural and technological changes in the environment and to monitor and predict the state of the solid Earth, the oceans and their living resources, the atmosphere, and the space environment of the Earth.

The major components of NOAA regularly produce various types of scientific and technical information in the following kinds of publications:

PROFESSIONAL PAPERS — Important definitive research results, major techniques, and special investigations.

CONTRACT AND GRANT REPORTS — Reports prepared by contractors or grantees under NOAA sponsorship.

ATLAS — Presentation of analyzed data generally in the form of maps showing distribution of rainfall, chemical and physical conditions of oceans and atmosphere, distribution of fishes and marine mammals, ionospheric conditions, etc.

TECHNICAL SERVICE PUBLICATIONS — Reports containing data, observations, instructions, etc. A partial listing includes data serials; prediction and outlook periodicals; technical manuals, training papers, planning reports, and information serials; and miscellaneous technical publications.

TECHNICAL REPORTS — Journal quality with extensive details, mathematical developments, or data listings.

TECHNICAL MEMORANDUMS — Reports of preliminary, partial, or negative research or technology results, interim instructions, and the like.

Information on availability of NOAA publications can be obtained from:

ENVIRONMENTAL SCIENCE INFORMATION CENTER
ENVIRONMENTAL DATA AND INFORMATION SERVICE
NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION
U.S. DEPARTMENT OF COMMERCE

Rockville, MD 20852